Discrete Choice Modeling
William Greene
Stern School of Business, New York University

Lab Session 9
Latent Class and Random Parameter Models

The first exercise is based on the conditional logit, mode choice data, clogit.lpj data

1. **Multinomial probit model**. Do the multinomial logit and multinomial probit models give similar results? You can't tell directly from the coefficient estimates because of scaling and normalization, so you have to rely on other indicators such as marginal effects. Fit a multinomial probit and a multinomial logit model, and compare the results. Note, estimation of the MNP model is ***extremely*** slow, so we have set it up with a very small number of replications and stopped the iterations at 25. This particular model would take 30-50 iterations to finish.

```
?
? 1. Multinomial Probit Model
?
NLOGIT          ; Lhs = Mode
                ; Choices = Air,Train,Bus,Car
                ; Rhs = TTME,INVC,INVT,GC; Rh2=One,Hinc
                ; Effects:GC(*) $
NLOGIT          ; Lhs = Mode ; MNP ; PTS = 10 ; Maxit = 25 ; Halton
                ; Choices = Air,Train,Bus,Car
                ; Rhs = TTME,INVC,INVT,GC; Rh2=One,Hinc
                ; Effects:GC(*) $
```

The next set of computations is based on the brand choices data, brandchoicesSP.lpj

Note that in these simulated data, the true underlying model really is a latent class data generating mechanishm, with three classes.

1. **Latent class model for brand choice**. First, fit a simple three class model with constant class probabilities. Then, fit the same model, but allow the class probabilities to very with age and sex. Finally, since we know that the true model is a three class model, we explore what happens when the model is over fit by fitting a four class model.

```
?
? (1) Basic 3 class model.
?
Nlogit ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None
    ; Rhs = Fash,Qual,Price,ASC4
    ; LCM ; Pds = 8 ; Pts = 3 $
?
? (2) 3 class model. Class probabilities depend on covariates
?
Nlogit ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None
    ; Rhs = Fash,Qual,Price,ASC4
    ; LCM=Male,Age25,Age39 ; Pds = 8 ; Pts = 3 $
?
? (3) Overspecified model.  4 class model.  The true model
?    underlying the data has three classes
Nlogit ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None
    ; Rhs = Fash,Qual,Price,ASC4
    ; LCM ; Pds = 8 ; Pts = 4 $
```

3. **Random parameters models**. We fit two specifications of a random parameters model. We also test the null hypothesis that the parameters are nonrandom.

```
?
? (4) Random parameters model
?
Nlogit ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None
      ; Rhs = Fash,Qual,Price,ASC4 $
CALC ; logl0 = logl $
Nlogit ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None
      ; Rhs = Fash,Qual,Price,ASC4
      ; RPL  ; Fcn= Fash(n),Price(n)
      ; Pds = 8 ; Pts = 25 $
CALC ; logl1 = logl $
CALC ; List ; chisq = 2*(logl1 – logl0) $
```

How many degrees of freedom are there for this test? Is the null hypothesis rejected?

```
?
? (5) Correlated parameters
?
Nlogit ; Lhs = Choice ; Choices=Brand1,Brand2,Brand3,None
      ; Rhs = Fash,Qual,Price,ASC4
      ; RPL  ; Fcn= Fash(n),Price(n) ; Correlated
      ; Pds = 8 ; Pts = 25 $
```

2. **Error Components logit model**. Fit the simple brand choice model with the addition of a person specific random effect. Note that here, we will take advantage of the fact that this is a panel. The same person is observed 8 times in each choice situation, so we assume that the effect does not change from one choice setting to the next. To speed this up, for purpose of the exercise, we use only 10 points in the simulation estimator. After obtaining the estimates, interpret your estimated model.

```
?
? (6)  An Error Components Logit model
?
ECLOGIT ; Lhs = Choice ; Choices = Brand1,Brand2,Brand3,None
      ; Rhs = Fash,Qual,Price,ASC4
      ; Pts = 10 ; Pds = 8
      ; ECM = (Brand1,Brand2,Brand3),(none) $
```